
Overview

This document lists TTCN changes needed to correct a problem in ATS ‘iwd-EUTRA-B2011-03_D11wk49’.

Corrections
Change 1 – Correction to PDN attach type when calling function ‘f_CDMA2000_RegisterHRPDCell()’

Function name
f_TC_6_2_2_3_CDMA2000 ()

Reason for change
As per TS 36.523-1 clause 6.2.2.3, it shall be checked in step 3 of the test procedure sequence whether the UE correctly sends an Access Probe to the HRPD cell after the E-UTRA cell has become non-suitable at “T2”. 

Note that according to the pre-test conditions, the UE is neither expected to be in HRPD dormant state nor having done a pre-registration to HRPD. Therefore, it cannot be expected from the UE to perform the call flow acc. to Core Spec TS 23.402 clause 9.4 (“E-UTRAN to HRPD mobility in idle-mode) which contains an Default Bearer Establishment together with the HRPD registration. Such state could only be reached in case of an Attach in due of a Handover from E-UTRA to HRPD (which would requires the UE to be in HRPD dormant state or to have performed a pre-registration to HRPD. 

So the only possibility to verify “with TTCN means” the successful reception of an Access Probe is to check if the HRPD Layer 2 Session Negotiation procedure was completed (( which equals an intermediate state “Registration with Session Open”; i.e. the UE informed about its capabilities but re-negotiation is still possible). Note that this does not equal a Default Bearer Establishment which would include completion of AKA procedure and VSNCP configuration procedure to allow for a subsequent PPP session.

Moreover, reference to ‘CDMA2000_Constants’ is missing in the import part of the module.



Summary of change
Function f_CDMA2000_RegisterHRPDCell() is called with non-default attach type ‘tsc_InitialAttach’ instead of ‘tsc_HandoverAttach’.

TTCN module
InterRat_CellReSelection_EtoC_CDMA2000.ttcn 

MCC160 Comment


Before:

...

module InterRat_CellReSelection_EtoC_CDMA2000 {
  /////////////////
  // Import Part //
  /////////////////
  import from CommonIratDefs all;
  import from CommonDefs all;
  import from CDMA2000_CommonFunctions all;
  import from CDMA2000_Component all;
  import from CDMA2000_CommonDefs all;
  //================================================================
  //
  // LTE -> C2K HRPD
  //
  //================================================================
  //================================================================
  /*
   * @desc      REFERENCE TS 36.523-1 clause 6.2.2.3
   * @status    APPROVED
   */
  function f_TC_6_2_2_3_CDMA2000 () runs on CDMA2000_PTC
  { /* Inter-RAT Cell selection / From E-UTRA RRC_IDLE to HRPD Idle / Serving cell becomes non-suitable (SServingCell<0) */
    const  PowerIor_Type tsc_Cell15_Ior := -60; //@sic R5-104718 sic@
    //Initialise test case
    f_CDMA2000_Init ( EUTRA_CDMA2000 );
    // Set MAX reference power corresponding to Ior/Ioc =-5 and -20 for cell 15 and 16 respectively
    // Note as cell 15 and 16 are on different frequency the
    f_CDMA2000_CellInfo_InitMaxReferencePower(cdma2000_Cell15, -55, tsc_Cell15_Ior,omit);    //Ioc =-55, Ior/Ioc=-5, Pilot offset =0
    //Create CDMA2000 cell15 and broadcast System Infomation
    f_CDMA2000CreateCell ( cdma2000_Cell15 );
    //First Trigger: to Change cell power according to T0 after EUTRA Registration
    f_WaitForIRATCoOrd_Trigger (EUTRA);
    f_CDMA2000_TestBody_Set( true );
    //------------------------------------------------
    //Start TestBody
    //@siclog "Step 1" siclog@
    f_CDMA2000_SetCellPower( cdma2000_Cell15, tsc_Cell15_Ior );
    //@siclog "Step 3" siclog@
    f_CDMA2000_RegisterHRPDCell (cdma2000_Cell15);
    f_CDMA2000_PreliminaryPass(__FILE__, __LINE__, "UE has sent the Access probe and registered on HRPD cell");
    //End TestBody
    //------------------------------------------------
    f_CDMA2000_TestBody_Set ( false );
    // send coordination message for end of test case
    f_SendIRATCoOrd ( EUTRA, cms_IRAT_Trigger );
    // Swith off/power off and release cells
    f_HRPD_Postamble (cdma2000_Cell15);    // UE is active and connected state
  }; //end of testcase 6.2.2.3
...
 

After:

...

module InterRat_CellReSelection_EtoC_CDMA2000 {
  /////////////////
  // Import Part //
  /////////////////
  import from CommonIratDefs all;
  import from CommonDefs all;
  import from CDMA2000_CommonFunctions all;
  import from CDMA2000_Component all;
  import from CDMA2000_CommonDefs all;
  import from CDMA2000_Constants all;
  //================================================================
  //
  // LTE -> C2K HRPD
  //
  //================================================================
  //================================================================
  /*
   * @desc      REFERENCE TS 36.523-1 clause 6.2.2.3
   * @status    APPROVED
   */
  function f_TC_6_2_2_3_CDMA2000 () runs on CDMA2000_PTC
  { /* Inter-RAT Cell selection / From E-UTRA RRC_IDLE to HRPD Idle / Serving cell becomes non-suitable (SServingCell<0) */
    const  PowerIor_Type tsc_Cell15_Ior := -60; //@sic R5-104718 sic@
    //Initialise test case
    f_CDMA2000_Init ( EUTRA_CDMA2000 );
    // Set MAX reference power corresponding to Ior/Ioc =-5 and -20 for cell 15 and 16 respectively
    // Note as cell 15 and 16 are on different frequency the
    f_CDMA2000_CellInfo_InitMaxReferencePower(cdma2000_Cell15, -55, tsc_Cell15_Ior,omit);    //Ioc =-55, Ior/Ioc=-5, Pilot offset =0
    //Create CDMA2000 cell15 and broadcast System Infomation
    f_CDMA2000CreateCell ( cdma2000_Cell15 );
    //First Trigger: to Change cell power according to T0 after EUTRA Registration
    f_WaitForIRATCoOrd_Trigger (EUTRA);
    f_CDMA2000_TestBody_Set( true );
    //------------------------------------------------
    //Start TestBody
    //@siclog "Step 1" siclog@
    f_CDMA2000_SetCellPower( cdma2000_Cell15, tsc_Cell15_Ior );
    //@siclog "Step 3" siclog@
    f_CDMA2000_RegisterHRPDCell (cdma2000_Cell15, -, -, tsc_InitialAttach);
    f_CDMA2000_PreliminaryPass(__FILE__, __LINE__, "UE has sent the Access probe and registered on HRPD cell");
    //End TestBody
    //------------------------------------------------
    f_CDMA2000_TestBody_Set ( false );
    // send coordination message for end of test case
    f_SendIRATCoOrd ( EUTRA, cms_IRAT_Trigger );
    // Swith off/power off and release cells
    f_HRPD_Postamble (cdma2000_Cell15);    // UE is active and connected state
  }; //end of testcase 6.2.2.3
...

Change 2 – Addition of new constant ‘tsc_InitialAttach’ 

Constant name
O3_Type

Reason for change
See change 1.

Summary of change
The new constant 

const O3_Type             tsc_InitialAttach := int2oct(1, 3);                                 

is added.

TTCN module
CDMA2000_Constants.ttcn

MCC160 Comment


Before:

...

module CDMA2000_Constants {
  import from CommonDefs all;
  import from CDMA2000_CommonDefs all;
  import from CDMA2000_ASP_TypeDefs all;
  const PhysCellIdCDMA2000_Type     tsc_PNOffsetDef := 50;              /* Arbitrarily Selected */
  const PhysCellIdCDMA2000_Type     tsc_PNOffset2 := 100;               /* Arbitrarily Selected  */
  const PhysCellIdCDMA2000_Type     tsc_PNOffset3 := 150;               /* Arbitrarily Selected  */
  const PhysCellIdCDMA2000_Type     tsc_PNOffset4 := 200;               /* Arbitrarily Selected  */
  const SearchWindowSize_Type       tsc_SearchWindowDef := 8;          /* @sic R5-115761 sic@ */
  const SystemType_Type             tsc_SystemTypeDef  := 0;                                            // Arbitrarily Selected
  const B8_Type                     tsc_SubNetMaskDef  := oct2bit('40'O);                               // Arbitrarily Selected
  //@sic R5s100848 sic@
  const MCC_Type                    tsc_CountryCodeDef := { 0, 0, 1 };                                  // default home PLMN MCC
  const MNC_Type                    tsc_MNC_Def := {  0, 1 };                                         // default home PLMN MNC
  const ReverseRateLimit_Type       tsc_ReverseRateLimitDef := kbps153_6;                               // Arbitrarily Selected
  const ReverseLinkMACIndex_Type    tsc_MACIndexDef    := 15;                                           // Arbitrarily selected
  const PacketApplication_Type      tsc_PacketAppDef   := enhMultiFlowPacketApp;                        // Arbitrarily Selected
  const ControlChannelRate_Type     tsc_ControlChannelRateDef := macIndex2;                             // Arbitrarily Selected
  const PilotOffset_Type            tsc_PilotOffsetDef := -7;  // Used for 1XRTT
  const PilotOffset_Type            tsc_PilotOffsetHRPDDef := 0; // In HRPD pilot is time multiplexed hence ofset is 0
                                                                //@sic R5-104718 sic@
  const PowerIor_Type               tsc_NonSuitableOffCellPowerIor := -121;                             // Corresponding to off cell
  const PowerIor_Type               tsc_PowerIorDef    := -55;                                          // Corresponding to Ior/Ioc =0 and Ioc =-55
  const PowerIoc_Type               tsc_PowerIocDef    := -55;
  const Attenuation_Type            tsc_AttenuationDef := 0;
  const O3_Type                     tsc_HandoverAttach := int2oct(3, 3);                                // 3: Handover attach to a PDN.
  const O3_Type                     tsc_UATI24         := '123456'O; // Arbitrarily selected
  const PDN_Id_Type                 tsc_PDN_Id         := '0000'B;  // Indicates the first PDN association;
  const RLP_FlowId_Type             tsc_RLP_FlowId1    := 1; // corresponding to DRB ID 1
  const float                       tsc_WatchDogDef    := 60.0; // 60 seconds to allow UE to complete the procedure on CDMA2000 cell
  const HRPD_UE_InitStateType       tsc_HRPD_UE_InitState_Active := active_Connected ; // UE is in PPP Active state and Radio access connected
  const LCP_DetachInit_Type         tsc_UEInitDetach  :=   UEInitiated  ; // Suitable at switch of UE
  // The dummy values defined here are to pass in the CoOrd_CDMA2000SysInfo_Type
  const SectorID_HRPD_Type          tsc_HRPD_SectorID_Dummy     := oct2bit('00000000000000000000000000000000'O);
  // Will be used in CoOrd_CDMA2000SysInfo_Type when cell is a RTT1X cell
  const B16_Type    tsc_BaseId_Dummy :='0000000000000000'B;
  // Will be used in CoOrd_CDMA2000SysInfo_Type when cell is a HRPD cell
  const B16_Type    tsc_NID_Dummy   :='0000000000000000'B ;
  // Will be used in CoOrd_CDMA2000SysInfo_Type when cell is a HRPD cell
  const B15_Type    tsc_SID_Dummy   :='000000000000000'B ;
  // Will be used in CoOrd_CDMA2000SysInfo_Type when cell is a HRPD cell
  const B12_Type    tsc_RegZone_Dummy   :='000000000000'B ;
  // Will be used in CoOrd_CDMA2000SysInfo_Type when cell is a HRPD cell
  const ColorCode_Type  tsc_ColorCode_Dummy := 0;
   // Will be used in CoOrd_CDMA2000SysInfo_Type when cell is a RTT1X cell
  const B12_Type tsc_Reg_ZoneDef := int2bit(1,12); // Registration zone 1
  const B12_Type tsc_Reg_Zone2 := int2bit(2,12); // Registration zone 1
  const B12_Type tsc_Reg_Zone3 := int2bit(3,12); // Registration zone 1
  const B12_Type tsc_Reg_Zone4 := int2bit(4,12); // Registration zone 1
  const B4_Type  tsc_BaseClassDef := int2bit (0,4); //Base class when Band class is not 1 and 4
  const B4_Type  tsc_BaseClass1_4 := int2bit (1,4); //Base class when Band class is  1 0r 4
  const TMSI_Zone_Type tsc_TMSI_ZoneDef := {int2bit(1,8)}; //Default TMSI Zone
  const ProtRev_Type  tsc_ProtRevDef := 6;   //Default Protocol revision
}// end of module
...
 

After:

...

module CDMA2000_Constants {
  import from CommonDefs all;
  import from CDMA2000_CommonDefs all;
  import from CDMA2000_ASP_TypeDefs all;
  const PhysCellIdCDMA2000_Type     tsc_PNOffsetDef := 50;              /* Arbitrarily Selected */
  const PhysCellIdCDMA2000_Type     tsc_PNOffset2 := 100;               /* Arbitrarily Selected  */
  const PhysCellIdCDMA2000_Type     tsc_PNOffset3 := 150;               /* Arbitrarily Selected  */
  const PhysCellIdCDMA2000_Type     tsc_PNOffset4 := 200;               /* Arbitrarily Selected  */
  const SearchWindowSize_Type       tsc_SearchWindowDef := 8;          /* @sic R5-115761 sic@ */
  const SystemType_Type             tsc_SystemTypeDef  := 0;                                            // Arbitrarily Selected
  const B8_Type                     tsc_SubNetMaskDef  := oct2bit('40'O);                               // Arbitrarily Selected
  //@sic R5s100848 sic@
  const MCC_Type                    tsc_CountryCodeDef := { 0, 0, 1 };                                  // default home PLMN MCC
  const MNC_Type                    tsc_MNC_Def := {  0, 1 };                                         // default home PLMN MNC
  const ReverseRateLimit_Type       tsc_ReverseRateLimitDef := kbps153_6;                               // Arbitrarily Selected
  const ReverseLinkMACIndex_Type    tsc_MACIndexDef    := 15;                                           // Arbitrarily selected
  const PacketApplication_Type      tsc_PacketAppDef   := enhMultiFlowPacketApp;                        // Arbitrarily Selected
  const ControlChannelRate_Type     tsc_ControlChannelRateDef := macIndex2;                             // Arbitrarily Selected
  const PilotOffset_Type            tsc_PilotOffsetDef := -7;  // Used for 1XRTT
  const PilotOffset_Type            tsc_PilotOffsetHRPDDef := 0; // In HRPD pilot is time multiplexed hence ofset is 0
                                                                //@sic R5-104718 sic@
  const PowerIor_Type               tsc_NonSuitableOffCellPowerIor := -121;                             // Corresponding to off cell
  const PowerIor_Type               tsc_PowerIorDef    := -55;                                          // Corresponding to Ior/Ioc =0 and Ioc =-55
  const PowerIoc_Type               tsc_PowerIocDef    := -55;
  const Attenuation_Type            tsc_AttenuationDef := 0;
const O3_Type                     tsc_HandoverAttach := int2oct(3, 3);                                // 3: Handover attach to a PDN.

  const O3_Type                     tsc_InitialAttach := int2oct(1, 3);                                // 1: Initial attach to a PDN (no pre-registration and not HRPD dormant).
  const O3_Type                     tsc_UATI24         := '123456'O; // Arbitrarily selected
  const PDN_Id_Type                 tsc_PDN_Id         := '0000'B;  // Indicates the first PDN association;
  const RLP_FlowId_Type             tsc_RLP_FlowId1    := 1; // corresponding to DRB ID 1
  const float                       tsc_WatchDogDef    := 60.0; // 60 seconds to allow UE to complete the procedure on CDMA2000 cell
  const HRPD_UE_InitStateType       tsc_HRPD_UE_InitState_Active := active_Connected ; // UE is in PPP Active state and Radio access connected
  const LCP_DetachInit_Type         tsc_UEInitDetach  :=   UEInitiated  ; // Suitable at switch of UE
  // The dummy values defined here are to pass in the CoOrd_CDMA2000SysInfo_Type
  const SectorID_HRPD_Type          tsc_HRPD_SectorID_Dummy     := oct2bit('00000000000000000000000000000000'O);
  // Will be used in CoOrd_CDMA2000SysInfo_Type when cell is a RTT1X cell
  const B16_Type    tsc_BaseId_Dummy :='0000000000000000'B;
  // Will be used in CoOrd_CDMA2000SysInfo_Type when cell is a HRPD cell
  const B16_Type    tsc_NID_Dummy   :='0000000000000000'B ;
  // Will be used in CoOrd_CDMA2000SysInfo_Type when cell is a HRPD cell
  const B15_Type    tsc_SID_Dummy   :='000000000000000'B ;
  // Will be used in CoOrd_CDMA2000SysInfo_Type when cell is a HRPD cell
  const B12_Type    tsc_RegZone_Dummy   :='000000000000'B ;
  // Will be used in CoOrd_CDMA2000SysInfo_Type when cell is a HRPD cell
  const ColorCode_Type  tsc_ColorCode_Dummy := 0;
   // Will be used in CoOrd_CDMA2000SysInfo_Type when cell is a RTT1X cell
  const B12_Type tsc_Reg_ZoneDef := int2bit(1,12); // Registration zone 1
  const B12_Type tsc_Reg_Zone2 := int2bit(2,12); // Registration zone 1
  const B12_Type tsc_Reg_Zone3 := int2bit(3,12); // Registration zone 1
  const B12_Type tsc_Reg_Zone4 := int2bit(4,12); // Registration zone 1
  const B4_Type  tsc_BaseClassDef := int2bit (0,4); //Base class when Band class is not 1 and 4
  const B4_Type  tsc_BaseClass1_4 := int2bit (1,4); //Base class when Band class is  1 0r 4
  const TMSI_Zone_Type tsc_TMSI_ZoneDef := {int2bit(1,8)}; //Default TMSI Zone
  const ProtRev_Type  tsc_ProtRevDef := 6;   //Default Protocol revision
}// end of module
...

Change 3 – Addition of new template to check for completion of HRPD Layer 2 Session Negotiation

Template name
RegAndDefBearerEstInd_Type

Reason for change
1. As per the above reasoning in change 1, a new template needs to be defined to check whether the HRPD Layer 2 Session Negotiation procedure is completed (= intermediate state “Registration with Session Open” ). This proves that an Access Probe was successfully received. 

Note that SessionNegotiationCmpl   can be misunderstood as related to a PPP Session instead of the HRPD Layer 2 Session. 

2. Minor editorial corrections required to some comments.

Summary of change
1. New template  cr_RegAndSessionOpenInd. is defined. 

Clarification is provided to the comment line for  SessionNegotiationCmpl  .

2. Minor editorial corrections are applied to some comments.

TTCN module
CDMA2000_Templates.ttcn 

MCC160 Comment


Before:

...

template RegAndDefBearerEstInd_Type cr_RegAndDefBearerEstInd_Def :=
  { /* @status    APPROVED */
    UATI_AssignmentCmpl     := true,  // UATI Assignment procedure is completed
    InitialChAssignCmpl     := true,  // Channel Assignment procedure is completed
    SCP_ConfigCmpl          := true,  // Session configuration procedure is completed
    Stream_ConfigCmpl       := true,  // Stream configuration procedure is completed
    EMPA_MMPA_ConfigCmpl    := true,  // Eanhanced multi flow packet application procedure is completed
    SessionNegotiationCmpl  := *,     // Session negotiation  procedure is optionally completed
    DeviceAuthCmpl          := *,     // Device authectication procedure is optionally completed
    LocationUpdateCmpl      := *,     // Location update procedure is optionally completed
    EAP_AKA_Cmpl            := true,  // AKA procedure is completed
    VSNCP_ConfigCmpl        := true,  // VSNCP  procedure is completed
    DHCP_ConfigCmpl         := *      // DHCP IPv4 address allocatione is optionally completed
  };
...
 

After:

...

template RegAndDefBearerEstInd_Type cr_RegAndDefBearerEstInd_Def :=
  { /* @status    APPROVED */
    UATI_AssignmentCmpl     := true,  // UATI Assignment procedure is completed
    InitialChAssignCmpl     := true,  // Channel Assignment procedure is completed
    SCP_ConfigCmpl          := true,  // Session configuration procedure is completed
    Stream_ConfigCmpl       := true,  // Stream configuration procedure is completed
    EMPA_MMPA_ConfigCmpl    := true,  // Enhanced multi flow packet application procedure is completed
    SessionNegotiationCmpl  := *,     // Layer 2 session negotiation procedure is optionally completed
    DeviceAuthCmpl          := *,     // Device authectication procedure is optionally completed
    LocationUpdateCmpl      := *,     // Location update procedure is optionally completed
    EAP_AKA_Cmpl            := true,  // AKA procedure is completed
    VSNCP_ConfigCmpl        := true,  // VSNCP  procedure is completed
    DHCP_ConfigCmpl         := *      // DHCP IPv4 address allocatione is optionally completed
  };
template RegAndDefBearerEstInd_Type cr_RegAndSessionOpenInd :=
    UATI_AssignmentCmpl     := true,  // UATI Assignment procedure is completed
    InitialChAssignCmpl     := true,  // Channel Assignment procedure is completed
    SCP_ConfigCmpl          := true,  // Session configuration procedure is completed
    Stream_ConfigCmpl       := true,  // Stream configuration procedure is completed
    EMPA_MMPA_ConfigCmpl    := true,  // Enhanced multi flow packet application procedure is completed
    SessionNegotiationCmpl  := true,  // Layer 2 session negotiation procedure is completed
    DeviceAuthCmpl          := *,     // Device authectication procedure is optionally completed
    LocationUpdateCmpl      := *,     // Location update procedure is optionally completed
    EAP_AKA_Cmpl            := ?,     // AKA procedure is not completed
    VSNCP_ConfigCmpl        := ?,     // VSNCP  procedure is not completed
    DHCP_ConfigCmpl         := *      // DHCP IPv4 address allocatione is optionally completed
};
...

Change 4 – Correction to Registration on HRPD cell

Function name
f_CDMA2000_RegisterHRPDCell ()

Reason for change
See change 1 and 3.

Summary of change
In case that the UE initial state is not pre-registered, then matching of the received SYSCMD_IND is done against the new non-default  cr_RegAndSessionOpenInd  template.

TTCN module
CDMA2000_CommonFunctions.ttcn 

MCC160 Comment


Before:

...

  //----------------------------------------------------------------------------
  /*
   * @desc      Performs a Registration procedure on an HRPD cell as defined in clause 10.4.1.1 of 36.523-3
   * @param     p_CellId
   * @param     p_RLP_FlowId        (default value: tsc_RLP_FlowId1)
   * @param     p_HRPD_UE_InitState (default value: idle_Null)
   * @param     p_CDMA2000_AttachType (default value: tsc_HandoverAttach)
   * @status    APPROVED
   */
  function f_CDMA2000_RegisterHRPDCell (CDMA2000_CellId_Type p_CellId,
                                        RLP_FlowId_Type p_RLP_FlowId := tsc_RLP_FlowId1,
                                        HRPD_UE_InitStateType p_HRPD_UE_InitState := idle_Null,
                                        CDMA2000_AttachType p_CDMA2000_AttachType := tsc_HandoverAttach) runs on CDMA2000_PTC
  {
    timer t_Watchdog := tsc_WatchDogDef;              // Local wait Timer
    t_Watchdog.start;
    SYSCMD_IND.send (cas_HRPD_SYSTEM_CMD(p_CellId,
                                         -, // timing info now
                                         cs_HRPD_SysCmd_RegAndDefBearerEst_Def(p_RLP_FlowId, p_HRPD_UE_InitState, p_CDMA2000_AttachType)));
    if (not (p_HRPD_UE_InitState == preregister))
      {// expect InitialAccessProb if not normal registration by a UE camping in HRPD cell
        SYSCMD_IND.receive(car_HRPD_SYSTEM_IND (p_CellId,
                                                -, // routing info none
                                                -, // timing info any
                                                cr_HRPD_SysInd_InitialAccesProbRcvd));
      }
    SYSCMD_IND.receive(car_HRPD_SYSTEM_IND (p_CellId,
                                            -, // routing info none
                                            -, // timing info any
                                            cr_SysInd_RegAndDefBearer(cr_RegAndDefBearerEstInd_Def)));
    t_Watchdog.stop;
  }
...
 

After:

...

  //----------------------------------------------------------------------------
  /*
   * @desc      Performs a Registration procedure on an HRPD cell as defined in clause 10.4.1.1 of 36.523-3
   * @param     p_CellId
   * @param     p_RLP_FlowId        (default value: tsc_RLP_FlowId1)
   * @param     p_HRPD_UE_InitState (default value: idle_Null)
   * @param     p_CDMA2000_AttachType (default value: tsc_HandoverAttach)
   * @status    APPROVED
   */
  function f_CDMA2000_RegisterHRPDCell (CDMA2000_CellId_Type p_CellId,
                                        RLP_FlowId_Type p_RLP_FlowId := tsc_RLP_FlowId1,
                                        HRPD_UE_InitStateType p_HRPD_UE_InitState := idle_Null,
                                        CDMA2000_AttachType p_CDMA2000_AttachType := tsc_HandoverAttach) runs on CDMA2000_PTC
  {
    timer t_Watchdog := tsc_WatchDogDef;              // Local wait Timer
    t_Watchdog.start;
    SYSCMD_IND.send (cas_HRPD_SYSTEM_CMD(p_CellId,
                                         -, // timing info now
                                         cs_HRPD_SysCmd_RegAndDefBearerEst_Def(p_RLP_FlowId, p_HRPD_UE_InitState, p_CDMA2000_AttachType)));
    if (not (p_HRPD_UE_InitState == preregister))
      {// expect InitialAccessProb if not normal registration by a UE camping in HRPD cell
        SYSCMD_IND.receive(car_HRPD_SYSTEM_IND (p_CellId,
                                                -, // routing info none
                                                -, // timing info any
                                                cr_HRPD_SysInd_InitialAccesProbRcvd));
        SYSCMD_IND.receive(car_HRPD_SYSTEM_IND (p_CellId,
                                            -, // routing info none
                                            -, // timing info any
                                            cr_SysInd_RegAndDefBearer(cr_RegAndSessionOpenInd)));
        t_Watchdog.stop;
    }
    else
      {
        SYSCMD_IND.receive(car_HRPD_SYSTEM_IND (p_CellId,
                                            -, // routing info none
                                            -, // timing info any
                                            cr_SysInd_RegAndDefBearer(cr_RegAndDefBearerEstInd_Def)));
        t_Watchdog.stop;
    }
}  

...

