
ASP Change

Before:

  type record TimeDomainRestriction_Type {

    MeasGapConfig_Type          MeasGapConfig                   /* measurement gap configuration acc. to TS 36.331, clause 6.3.5 and gap pattern acc. TS 36.133 Table 8.1.2.1-1 */

  };

  type record CcchDcchDtchConfig_Type {

    TimeDomainRestriction_Type  TimeDomainRestriction optional, /* to tell the SS when no assignments/grants shall be assigned to the UE */

    CcchDcchDtchConfigDL_Type   DL                    optional, /* Scheduling, parameters related to CCCH, DCCH and DTCH in DL */

    CcchDcchDtchConfigUL_Type   UL                    optional, /* Scheduling, parameters related to CCCH, DCCH and DTCH in UL */

    DrxCtrl_Type                DrxCtrl               optional, /* DRX configuration as sent to the UE (or 'None' when the UE does not support connected mode DRX) */

    TTI_BundlingConfig_Type     TtiBundling           optional  /* TTI bundling as configured at the UE */

  };

After:

  type union MeasGapCtrl_Type {                                 /* support of measurement gap configuration !!!! MEASGAPCONFIG !!!!: to replace TimeDomainRestriction_Type */

    Null_Type                   None,                           /* no measurement gap configuration  */

    MeasGapConfig_Type          Config                          /* measurement gap configuration acc. to TS 36.331, clause 6.3.5 and gap pattern acc. TS 36.133 Table 8.1.2.1-1;

                                                                   NOTE: the release branch of MeasGapConfig in general is not used for configuration of the SS */

  };

  type record CcchDcchDtchConfig_Type {

    MeasGapCtrl_Type            MeasGapCtrl           optional, /* to tell the SS when no assignments/grants shall be assigned to the UE */

    CcchDcchDtchConfigDL_Type   DL                    optional, /* Scheduling, parameters related to CCCH, DCCH and DTCH in DL */

    CcchDcchDtchConfigUL_Type   UL                    optional, /* Scheduling, parameters related to CCCH, DCCH and DTCH in UL */

    DrxCtrl_Type                DrxCtrl               optional, /* DRX configuration as sent to the UE (or 'None' when the UE does not support connected mode DRX) */

    TTI_BundlingConfig_Type     TtiBundling           optional  /* TTI bundling as configured at the UE */

  };

( Initialisation:

  template (value) SYSTEM_CTRL_REQ cas_CellConfig_Def_REQ(CellId_Type                              p_CellId,

                                                          template (value) TimingInfo_Type         p_TimingInfo,

                                                          template (value) StaticCellInfo_Type     p_StaticCellInfo,

                                                          template (value) AllSiSchedul_Type       p_SiSchedul,

                                                          EUTRA_CellInfo_Type     p_CellInfo   ) :=

  { /* @status    APPROVED */

    /* @sic R5w110104: MimoInfo removed from CcchDcchDtchConfigDL_Type sic@ */

    Common := cs_ReqAspCommonPart_CellCfg(p_CellId, p_TimingInfo),

    Request := {

      Cell := {

        AddOrReconfigure := {

          Basic := {

            ConfigCapability := p_CellInfo.CellConfigCapability,

            StaticCellInfo := p_StaticCellInfo,

            PhysicalLayerConfigDL := cs_PhysicalLayerConfigDL (p_CellInfo.FDD_TDD_Info.FDD_TDD,

                                                               p_CellInfo.Frequency.DL_ChBandwidth,

                                                               cs_PHICH_Config_r8(p_CellInfo.Sysinfo.BCCH_Info.MIB.message_.phich_Config),

                                                               p_CellInfo.PhysicalParameters.Cfi,

                                                               p_CellInfo.AntennaInfo.SS_AntennaInfoCommon),

            InitialCellPower := {

              MaxReferencePower := p_CellInfo.ReferencePower.MaxReferencePower,

              Attenuation := p_CellInfo.ReferencePower.Attenuation

            },

            BcchConfig := {

              //Sib2Config := cs_BCCH_Configuration,

              Pbch := true,                                 /* no configuration parameters required */

              Pdsch := {

                Sib1Schedul := {

                  DciInfo := cs_DciInfo_Sib1

                },

                SiSchedul := p_SiSchedul

              },

              BcchInfo := p_CellInfo.Sysinfo.BCCH_Info

            },

            PcchConfig := {

              DciInfo := cs_DciInfo_Pcch (p_CellInfo.Frequency.DL_ChBandwidth)

            }

          },

          Active := {

            C_RNTI := p_CellInfo.RNTI.C_RNTI,

            PhysicalLayerConfigUL := {

              Prach := cs_PRACH_Config_r8( cs_PRACH_Config_Def( p_CellInfo.Sysinfo.BCCH_Info.SIs[0].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib2.radioResourceConfigCommon.prach_Config) ),

              Pucch := {

                Common := cs_PUCCH_ConfigCommon_r8( p_CellInfo.Sysinfo.BCCH_Info.SIs[0].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib2.radioResourceConfigCommon.pucch_ConfigCommon ),      /* @sic R5s100135 sic@ */

                Dedicated := cs_PUCCH_ConfigDedicated_r8( p_CellInfo.FDD_TDD_Info.pucch_Config )

              },

              Pusch := {

                Common := cs_PUSCH_ConfigCommon_r8( p_CellInfo.Sysinfo.BCCH_Info.SIs[0].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib2.radioResourceConfigCommon.pusch_ConfigCommon ),     /* @sic R5s100135 sic@ */

                Dedicated := cs_PUSCH_ConfigDedicated_r8( cs_508_PUSCH_ConfigDedicated_Default )

              },

              TimingAdvance := {

                InitialValue := 0

              },

              SRS_UL_Config := {

                Common := cs_SoundingRS_UL_ConfigCommon_r8( p_CellInfo.Sysinfo.BCCH_Info.SIs[0].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib2.radioResourceConfigCommon.soundingRS_UL_ConfigCommon ),     /* @sic R5s100135 sic@ */

                Dedicated := cs_SoundingRS_UL_ConfigDedicated_r8( p_CellInfo.FDD_TDD_Info.soundingRS_UL_Config )

              },

              SR_Config := cs_SchedulingRequestConfig_r8( cs_508_SchedulingRequest_Config_Default (p_CellInfo.PhysicalParameters.sr_PUCCH_ResourceIndex, p_CellInfo.FDD_TDD_Info.sr_ConfigIndex)),  // @sic R5s090312 R5-102177 sic@

              CQI_ReportConfig := cs_CQI_ReportConfig_r8andLater( cs_508_CQI_ReportConfig_Default(p_CellInfo.FDD_TDD_Info.cqi_ReportConfig) ),

              UplinkPowerControlCommon := cs_UplinkPowerControlCommon_r8(p_CellInfo.Sysinfo.BCCH_Info.SIs[0].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib2.radioResourceConfigCommon.uplinkPowerControlCommon),     /* @sic R5s100135 sic@ */

              UplinkPowerControlDedicated := cs_UplinkPowerControlDedicated_r8(cs_508_UplinkPowerControlDedicated_Default)

            },

            RachProcedureConfig := cs_RachProcedureConfig_Def( cs_RACH_ConfigCommon_r8(p_CellInfo.Sysinfo.BCCH_Info.SIs[0].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib2.radioResourceConfigCommon.rach_ConfigCommon), p_CellInfo.PhysicalParameters.RAR_TA, p_CellInfo.Frequency.DL_ChBandwidth ),     /* @sic R5s100135 sic@ */

            CcchDcchDtchConfig := {

              MeasGapConfig := cs_MeasGapCtrl_None,
              DL := {

                DciInfo := { Auto := cs_DciInfo_CcchDcchDtchDL(p_CellInfo.Frequency.DL_ChBandwidth)},

                AntennaInfo := cs_AntennaInfoDedicated_r8andLater( p_CellInfo.AntennaInfo.SS_AntennaInfoDedicated ),

                HarqProcessConfig := { /* @sic R5w100214, R5w100220 sic@ */

                  AllProcesses := true

                }

              },

              UL := {

                DciInfo := cs_DciInfo_CcchDcchDtchUL_Def,

                Hopping := {

                  Deactivated := true       // FFS

                },

                PUCCH_Synch  := { None  := true},

                UL_GrantConfig  := { None  := true}

              },

              DrxCtrl := cs_DrxCtrl_None, /* no connected mode DRX configured per default */

              TtiBundling := cs_TTI_BundlingConfig_r8(false)

            }

          }

        }

      }

    }

  };

New templates:

· cs_MeasGapCtrl_None
· cs_MeasGapCtrl_Config
· cas_MeasGapCtrl_REQ
New Cell Information: MeasGapCtrl

  type record EUTRA_CellInfo_Type {

    CellIdentities_Type            CellIds,

    Frequency_IE_Type              Frequency, 

    EUTRA_FDD_TDD_CellInfo_Type    FDD_TDD_Info,  

    EUTRA_CellSysInfo_Type         Sysinfo,

    NAS_Parameter_Type             NAS_Parameters,

    CellTimingInfo_Type            CellTimingInfo,

    ReferenceCellPower_Type        ReferencePower,

    CellConfigCapability_Type      CellConfigCapability, 

    RNTI_Type                      RNTI, 

    PhysicalParameters_Type        PhysicalParameters,

    AntennaInfo_Type               AntennaInfo,

    DrxCtrl_Type                   DrxCtrl,

    MeasGapCtrl_Type               MeasGapCtrl
  };

( Initialisation:

  template (value) EUTRA_CellInfo_Type cs_EUTRA_CellInfoDef(template (value) CellIdentities_Type     p_CellIds,

                                                            template (value) Combination_Type        p_Combination,

                                                            template (value) FrequencyBand_Type      p_FrequencyBand,

                                                            template (value) CarrierFreqEUTRA        p_UL_DL_Earfcn,

                                                            Dl_Bandwidth_Type       p_DL_ChBandwidth,

                                                            template (omit)  Ul_Bandwidth_Type       p_UL_ChBandwidth,

                                                            template (value) NAS_Parameter_Type      p_NAS_Param,

                                                            template (value) BcchInfo_Type           p_BCCH_Info,

                                                            template (value) CellTimingInfo_Type     p_CellTimingInfo,

                                                            template (value) EUTRA_FDD_TDD_CellInfo_Type   p_FDD_TDD_Info,

                                                            CfiValue_Type                            p_CfiValue,

                                                            RACH_TimingAdvance_Type                  p_RAR_TA,

                                                            SR_PUCCH_ResourceIndex_Type              p_SR_PUCCH_ResourceIndex   // @sic R5s090312 sic@

                                                            ) :=

  { /* @desc      Default cell info value

     * @status    APPROVED */

    CellIds      :=  p_CellIds,

    Frequency := {

      FrequencyBandIndicator :=  p_FrequencyBand,

      DL_ChBandwidth := p_DL_ChBandwidth,

      UL_ChBandwidth := p_UL_ChBandwidth,

      UL_DL_Earfcn           :=  p_UL_DL_Earfcn

    },

    FDD_TDD_Info := p_FDD_TDD_Info,

    Sysinfo := {

      BCCH_Info           :=  p_BCCH_Info,

      Combination         :=  p_Combination,

      SB1_ValueTagChanged :=  false

    },

    NAS_Parameters    :=  p_NAS_Param,

    CellTimingInfo    :=  p_CellTimingInfo,

    ReferencePower := {

      MaxReferencePower := tsc_ServingCellRS_EPRE,

      Attenuation := {Off := true}

    },

    CellConfigCapability := fullCell,         // all cells are intialised by default with fullCell

    RNTI  := { C_RNTI := tsc_C_RNTI_Def },

    PhysicalParameters := {

      Cfi := p_CfiValue,

      RAR_TA := p_RAR_TA,

      sr_PUCCH_ResourceIndex := p_SR_PUCCH_ResourceIndex,

      PDSCH_p_b := tsc_P_b_1Tx      // acc to 36.508 Table 4.6.3-5: PDSCH-ConfigCommon-DEFAULT @sic R5-100786 sic@

    },

    AntennaInfo := {  // default value for 1TX   @sic R5-100786 sic@

      SS_AntennaInfoCommon := cs_AntennaInfoCommon_1Port,

      AntennaInfoCommon := omit,

      SS_AntennaInfoDedicated := cs_AntennaInfoDedicated_Def,

      AntennaInfoDedicated := cs_PhysicalConfigDedicated_AntennaInfo_Def

    },

    DrxCtrl := {  // @sic R5-104796 DRX support sic@

      None := true

    },

    MeasGapCtrl := {
      None := true
    }
  };

New functions:

· f_EUTRA_CellInfo_GetMeasGapCtrl
· f_EUTRA_CellInfo_MeasGapIsConfigured
· f_EUTRA_CellInfo_SetMeasGapCtrl
· f_EUTRA_SS_MeasGapCtrlConfig
New Function

  function f_EUTRA_RRCConnectionReconfiguration_MeasConfig(CellId_Type p_CellId,

                                                           template (value) DL_DCCH_Message p_RRCConnectionReconfiguration) runs on EUTRA_PTC

  {

    var template (value) RRC_TransactionIdentifier v_RRC_TI := p_RRCConnectionReconfiguration.message_.c1.rrcConnectionReconfiguration.rrc_TransactionIdentifier;

    var template (value) MeasGapCtrl_Type v_MeasGapCtrl;

    var template (value) MeasGapConfig v_MeasGapConfig;

    var SubFrameTiming_Type v_Timing;

    if (ispresent(p_RRCConnectionReconfiguration.message_.c1.rrcConnectionReconfiguration.criticalExtensions.c1.rrcConnectionReconfiguration_r8.measConfig.measGapConfig)) {

      v_MeasGapConfig := p_RRCConnectionReconfiguration.message_.c1.rrcConnectionReconfiguration.criticalExtensions.c1.rrcConnectionReconfiguration_r8.measConfig.measGapConfig;

      if (ischosen(v_MeasGapConfig.setup)) {

        v_MeasGapCtrl := cs_MeasGapCtrl_Config(v_MeasGapConfig);

      } else {

        v_MeasGapCtrl := cs_MeasGapCtrl_None;

      }

      v_Timing := f_EUTRA_GetNextSendOccasion(p_CellId);

      SRB.send(cas_SRB1_RrcPdu_REQ(p_CellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number), p_RRCConnectionReconfiguration));

      v_Timing := f_EUTRA_TimingInfoAdd(v_Timing, 5);   // add 5ms

      f_EUTRA_SS_MeasGapCtrlConfig(p_CellId, v_MeasGapCtrl, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));

    }

    else {

      SRB.send(cas_SRB1_RrcPdu_REQ(p_CellId, cs_TimingInfo_Now, p_RRCConnectionReconfiguration));

    }

    SRB.receive(car_SRB1_RrcPdu_IND(p_CellId, cr_508_RRCConnectionReconfigurationComplete(v_RRC_TI)));

  }

This function shall be applied for all cases when measConfig is contained in the RRCConnectionReconfiguration, i.e. even when the MeasConfig does not contain any measGapConfig the function shall be used to be future proven.

In detail the function shall be applied for 

Module
Test case
Template
MeasGapConfig

MultiLayer_Procedures
TC_13_2_1
cs_RRCConnectionReconfiguration_13_2_1_Step3
omit

MultiLayer_Procedures
TC_13_4_2_1
cs_508_RRCConnectionReconfiguration_MeasConfig_UTRAN
cs_508_MeasGapConfig_GP1

MultiLayer_Procedures
TC_13_4_2_2
cs_RRCConnectionReconfiguration_13_4_2_2_Step3
cs_508_MeasGapConfig_GP2

RRC_Measurement
TC_8_3_1_9
cs_RRCConnectionReconfiguration_ResumeInterFreqMeas
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_1_10
cs_RRCConnectionReconfiguration_ResumeInterFreqMeas
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_1_11
cs_RRCConnectionReconfiguration_ResumeInterFreqMeas
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_1_1
cs_RRCConnectionReconfiguration_8_3_1_1_Step1
omit

RRC_Measurement
TC_8_3_1_2
cs_RRCConnectionReconfiguration_8_3_1_2_Step1
omit

RRC_Measurement
TC_8_3_1_3
cs_RRCConnectionReconfiguration_8_3_1_3_Step1
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_1_4
cs_RRCConnectionReconfiguration_8_3_1_4_Step1
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_1_4
cs_RRCConnectionReconfiguration_8_3_1_4_Step10
omit

RRC_Measurement
TC_8_3_1_5
cs_RRCConnectionReconfiguration_8_3_1_5_Step1
omit

RRC_Measurement
TC_8_3_1_6
cs_RRCConnectionReconfiguration_8_3_1_6_Step1
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_1_7
cs_RRCConnectionReconfiguration_8_3_1_7_Step1
omit

RRC_Measurement
TC_8_3_1_7
cs_RRCConnectionReconfiguration_8_3_1_7_Step7
omit

RRC_Measurement
TC_8_3_1_8
cs_RRCConnectionReconfiguration_8_3_1_8_Step1
omit

RRC_Measurement
TC_8_3_1_8
cs_MeasConfig_8_3_1_8_Step5
omit

RRC_Measurement
TC_8_3_1_9
cs_RRCConnectionReconfiguration_8_3_1_9_Step1_pc25
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_1_9
cs_RRCConnectionReconfiguration_8_3_1_9_Step1_notpc25
omit

RRC_Measurement
TC_8_3_1_10
cs_RRCConnectionReconfiguration_8_3_1_10_Step1
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_1_11
cs_RRCConnectionReconfiguration_8_3_1_11_Step1_pc25
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_1_11
cs_RRCConnectionReconfiguration_8_3_1_11_Step1_notpc25
omit

RRC_Measurement
TC_8_3_2_1
cs_RRCConnectionReconfiguration_8_3_2_1_Step1
cs_508_MeasGapConfig_GP2

RRC_Measurement
TC_8_3_2_2
cs_RRCConnectionReconfiguration_8_3_2_2_Step1
cs_508_MeasGapConfig_GP2

RRC_Measurement
TC_8_3_2_2
cs_RRCConnectionReconfiguration_RemoveMeasId_2Entries
cs_508_MeasGapConfig_GP2

RRC_Measurement
TC_8_3_2_5
cs_RRCConnectionReconfiguration_RemoveMeasId_2Entries
cs_508_MeasGapConfig_GP2

RRC_Measurement
TC_8_3_2_3
cs_RRCConnectionReconfiguration_8_3_2_3_Step1_FDD
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_2_3
cs_RRCConnectionReconfiguration_8_3_2_3_Step1_TDD
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_2_4
cs_RRCConnectionReconfiguration_8_3_2_4_Step1_FDD
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_2_4
cs_RRCConnectionReconfiguration_8_3_2_4_Step1_TDD
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_2_4
cs_RRCConnectionReconfiguration_RemoveMeasId_1Entry
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_2_5
cs_RRCConnectionReconfiguration_8_3_2_5_Step1_FDD
cs_508_MeasGapConfig_GP2

RRC_Measurement
TC_8_3_2_5
cs_RRCConnectionReconfiguration_8_3_2_5_Step1_TDD
cs_508_MeasGapConfig_GP2

RRC_Measurement
TC_8_3_2_6
cs_RRCConnectionReconfiguration_8_3_2_6_Step1_FDD
cs_508_MeasGapConfig_GP2

RRC_Measurement
TC_8_3_2_6
cs_RRCConnectionReconfiguration_8_3_2_6_Step1_TDD
cs_508_MeasGapConfig_GP2

RRC_Measurement
TC_8_3_3_1
cs_RRCConnectionReconfiguration_8_3_3_1_Step1
omit

RRC_Measurement
TC_8_3_3_1
cs_RRCConnectionReconfiguration_8_3_3_1_Step5
omit

RRC_Measurement
TC_8_3_3_2
cs_RRCConnectionReconfiguration_8_3_3_2_Step1_FDD
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_3_2
cs_RRCConnectionReconfiguration_8_3_3_2_Step1_TDD
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_3_2
cs_RRCConnectionReconfiguration_8_3_3_2_Step5_FDD
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_3_2
cs_RRCConnectionReconfiguration_8_3_3_2_Step5_TDD
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_3_3
cs_RRCConnectionReconfiguration_8_3_3_3_Step1
cs_508_MeasGapConfig_GP2

RRC_Measurement
TC_8_3_3_3
cs_RRCConnectionReconfiguration_8_3_3_3_Step5
cs_508_MeasGapConfig_GP2

RRC_Measurement
TC_8_3_4_2
cs_RRCConnectionReconfiguration_8_3_4_2_Step1
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_4_2
cs_RRCConnectionReconfiguration_8_3_4_2_Step5
cs_MeasGapConfig_release

RRC_Measurement
TC_8_3_4_3
cs_RRCConnectionReconfiguration_8_3_4_3_Step1
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_4_3
cs_RRCConnectionReconfiguration_8_3_4_3_Step5
cs_MeasGapConfig_release

RRC_Measurement
TC_8_3_4_4
cs_RRCConnectionReconfiguration_8_3_4_4_Step1_FDD
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_4_4
cs_RRCConnectionReconfiguration_8_3_4_4_Step1_TDD
cs_508_MeasGapConfig_GP1

RRC_Measurement
TC_8_3_4_4
cs_RRCConnectionReconfiguration_8_3_4_4_Step5_FDD
cs_MeasGapConfig_release

RRC_Measurement
TC_8_3_4_4
cs_RRCConnectionReconfiguration_8_3_4_4_Step5_TDD
cs_MeasGapConfig_release






RRC_Measurement_C2K
TC_8_3_3_4
cs_RRCConnectionReconfiguration_IratMeasCGI_C2K
cs_508_MeasGapConfig_GP2

RRC_Measurement_C2K
TC_8_3_3_5
cs_RRCConnectionReconfiguration_IratMeasCGI_C2K
cs_508_MeasGapConfig_GP2

RRC_Measurement_C2K
TC_8_3_2_8
cs_RRCConnectionReconfiguration_IratPerMeas_C2K
cs_508_MeasGapConfig_GP2

RRC_Measurement_C2K
TC_8_3_2_10
cs_RRCConnectionReconfiguration_IratPerMeas_C2K
cs_508_MeasGapConfig_GP2

RRC_Measurement_C2K
TC_8_3_2_8
cs_RRCConnectionReconfiguration_MeasRemove
omit

RRC_Measurement_C2K
TC_8_3_2_10
cs_RRCConnectionReconfiguration_MeasRemove
omit






RRC_Measurement_C2K
TC_8_3_2_7
cs_RRCConnectionReconfiguration_IratMeasB2_C2K
cs_508_MeasGapConfig_GP2

RRC_Measurement_C2K
TC_8_3_2_9
cs_RRCConnectionReconfiguration_IratMeasB2_C2K
cs_508_MeasGapConfig_GP2

RRC_Measurement_C2K
TC_8_3_3_4
cs_RRCConnectionReconfiguration_IratMeasB2_C2K
cs_508_MeasGapConfig_GP2

RRC_Measurement_C2K
TC_8_3_3_5
cs_RRCConnectionReconfiguration_IratMeasB2_C2K
cs_508_MeasGapConfig_GP2

InterRat_Handover_to_C2K
TC_8_4_5_4
cs_RRCConnectionReconfiguration_IratMeasB2_C2K
cs_508_MeasGapConfig_GP2

InterRat_Handover_to_C2K
TC_8_4_7_1
cs_RRCConnectionReconfiguration_IratMeasB2_C2K
cs_508_MeasGapConfig_GP2






InterRat_Handover_to_GERAN
TC_8_4_3_2 TC_8_4_3_3
cs_RRCConnectionReconfiguration_8_4_3_2and3_Step1
omit






MAC_712
TC_7_1_2_1
cs_Measurement_Config_A3
omit per default

RRC_Handover
TC_8_2_4_1
cs_Measurement_Config_A3
omit per default

RRC_Handover
TC_8_2_4_2
cs_Measurement_Config_A3
omit per default

RRC_Handover
TC_8_2_4_5
cs_Measurement_Config_A3
omit per default

RRC_Handover
TC_8_2_4_7
cs_Measurement_Config_A3
omit per default

RRC_Handover
TC_8_2_4_8
cs_Measurement_Config_A3
omit per default






MultiLayer_Procedures
TC_13_4_1_2
cs_Measurement_Config_A3_InterFreq
cs_508_MeasGapConfig_GP1

MultiLayer_Procedures
TC_13_4_1_3
(2 times)
cs_Measurement_Config_A3_InterFreq
cs_508_MeasGapConfig_GP1

RRC_Handover
TC_8_2_4_6
cs_Measurement_Config_A3_InterFreq
cs_508_MeasGapConfig_GP1






Further changes in TTCN:

· In many cases optional parameters are handed over to cs_MeasConfig with their default values; that is not needed and the list of parameters can be shortened

Changed Functions

f_EUTRA_SS_CommonCellConfig

  function f_EUTRA_SS_CommonCellConfig(CellId_Type p_CellId,

                                       template (value) SYSTEM_CTRL_REQ p_SYSTEM_CTRL_REQ) runs on EUTRA_PTC

  { /* @sic R5-104796 DRX support sic@

       Note: function updates DRX parameters in cell configuration; in the future the TA period may need to be taken into account too (FFS) */

    var CellConfigInfo_Type v_CellConfigInfo;

    var DrxCtrl_Type v_DrxCtrl;

    var MeasGapCtrl_Type v_MeasGapCtrl;

    if (not ischosen(p_SYSTEM_CTRL_REQ.Request.Cell)) {

      FatalError(__FILE__, __LINE__, "f_EUTRA_SS_CommonCellConfig shall get handed over CellConfigRequest");

    }

    if (ischosen(p_SYSTEM_CTRL_REQ.Request.Cell.AddOrReconfigure)) {

      v_CellConfigInfo := valueof(p_SYSTEM_CTRL_REQ.Request.Cell.AddOrReconfigure);

      if (ispresent(v_CellConfigInfo.Active)) {

        if (ispresent(v_CellConfigInfo.Active.CcchDcchDtchConfig)) {

          if (ispresent(v_CellConfigInfo.Active.CcchDcchDtchConfig.DrxCtrl)) {

            v_DrxCtrl := v_CellConfigInfo.Active.CcchDcchDtchConfig.DrxCtrl;

            f_EUTRA_CellInfo_SetDrxCtrl(p_CellId, v_DrxCtrl);

          }

          if (ispresent(v_CellConfigInfo.Active.CcchDcchDtchConfig.MeasGapConfig)) {
            v_MeasGapCtrl := v_CellConfigInfo.Active.CcchDcchDtchConfig.MeasGapConfig;
            f_EUTRA_CellInfo_SetMeasGapCtrl(p_CellId, v_MeasGapCtrl);
          }
        }

      }

    }

    SYS.send( p_SYSTEM_CTRL_REQ );

    if (valueof(p_SYSTEM_CTRL_REQ.Common.ControlInfo.CnfFlag) == true) {  /* @sic R5s100135 sic@ */

      SYS.receive ( car_CellConfig_CNF ( p_CellId ) );

    }

  }

f_EUTRA_GetNextSendOccasion

  function f_EUTRA_GetNextSendOccasion ( CellId_Type p_CellId,

                                         UInt_Type p_MilliSeconds := 100) runs on EUTRA_PTC return SubFrameTiming_Type

  {

    var SubFrameTiming_Type v_Timing;

    var SubFrameTiming_Type v_NextSendOccasion;

    var boolean v_MeasGapIsConfigured := f_EUTRA_CellInfo_MeasGapIsConfigured(p_CellId);
    var boolean v_DrxIsConfigured := f_EUTRA_CellInfo_DrxIsConfigured(p_CellId); 

    if (p_MilliSeconds < 100) {

      FatalError(__FILE__, __LINE__, "next send occasion shall be at least 100ms in advance");

    }

    if (v_MeasGapIsConfigured and v_DrxIsConfigured) {
      FatalError(__FILE__, __LINE__, "f_EUTRA_GetNextSendOccasion: no support for simultaneous configurtion of DRX and MeasGap");
    }
    v_Timing := f_EUTRA_GetCurrentTiming(p_CellId);

    v_NextSendOccasion := f_EUTRA_TimingInfoAdd(v_Timing, p_MilliSeconds);

    v_NextSendOccasion := f_EUTRA_TimingInfoAdjust(v_NextSendOccasion);

    if (v_DrxIsConfigured) {

      v_NextSendOccasion := f_EUTRA_GetDrxOnDuration(p_CellId, v_NextSendOccasion);

    }

    if (v_MeasGapIsConfigured) {
      v_NextSendOccasion := f_EUTRA_SkipMeasGap(p_CellId, v_NextSendOccasion);
    }
    return v_NextSendOccasion;

  }

New Function:

  function f_EUTRA_SkipMeasGap(CellId_Type p_CellId,

                               SubFrameTiming_Type p_SubFrameTiming) runs on EUTRA_PTC return SubFrameTiming_Type

  { /* calculation of measurment gap acc to 36.331 cl. 5.5.2.9 depending on configuration as defined in 36.133 Table 8.1.2.1-1 */

    var MeasGapCtrl_Type v_MeasGapCtrl := f_EUTRA_CellInfo_GetMeasGapCtrl(p_CellId);

    var SubFrameTiming_Type v_SubFrameTiming := p_SubFrameTiming;       // => per default the timing is not changed

    var MeasGapConfig.setup.gapOffset v_MeasGapOffset;

    var UInt_Type v_GapSfnPeriod;

    var UInt_Type v_GapOffset;

    var UInt_Type v_SfnBase;

    var UInt_Type v_GapStart;

    var UInt_Type v_GapEnd;

    var UInt_Type v_SFN;

    var UInt_Type v_Subframes;

    if (ischosen(v_MeasGapCtrl.Config.R8.setup)) {

      v_MeasGapOffset := v_MeasGapCtrl.Config.R8.setup.gapOffset;

      if (ischosen(v_MeasGapOffset.gp0)) {

        v_GapSfnPeriod := 4;

        v_GapOffset := v_MeasGapOffset.gp0;

      }

      else if (ischosen(v_MeasGapOffset.gp1)) {

        v_GapSfnPeriod := 8;

        v_GapOffset := v_MeasGapOffset.gp1;

      }

      else {

        FatalError(__FILE__, __LINE__, "f_EUTRA_SkipMeasGap: unknown MeasGap configuration");

      }

      v_SFN := p_SubFrameTiming.SFN.Number;

      v_Subframes := (v_SFN * 10) + p_SubFrameTiming.Subframe.Number;

      v_SfnBase := (v_SFN / v_GapSfnPeriod) * v_GapSfnPeriod;   // SFN rounded down to the next multiple of v_GapSfnPeriod

      v_GapStart := (v_SfnBase * 10) + v_GapOffset;             // start of the next gap in subframes

      v_GapEnd := v_GapStart + 6;                               // end of the next gap in subframes (acc. 36.133 Table 8.1.2.1-1 MeasurementGapLength is 6ms for bothe configurations)

      if ((v_GapStart <= v_Subframes) and (v_Subframes < v_GapEnd)) {

        v_SFN := (v_SFN + 1) mod 1024;

        v_SubFrameTiming.SFN.Number := v_SFN;

      }

    }

    return v_SubFrameTiming;

  }

f_EUTRA_508RRC_IntraLTE_HO_InterCell

  function f_EUTRA_508RRC_IntraLTE_HO_InterCell ( CellId_Type                           p_SourceCellId,

                                                  CellId_Type                           p_TargetCellId,

                                                  template (value) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_Def,

                                                  C_RNTI                                p_C_RNTI        := tsc_C_RNTI_Def2,

                                                  template (omit)  MeasConfig           p_MeasConfig    := omit,

                                                  template (omit)  DciUlInfo_Type       p_DciUlInfo_TargetCell      := cs_DciInfo_CcchDcchDtchUL_Def,

                                                  template (value) UL_GrantConfig_Type  p_UL_GrantConfig_SourceCell := cs_UL_GrantConfig_Def,

                                                  template (value) UL_GrantConfig_Type  p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_Def,

                                                  Dsr_TransMax_Type                     p_Dsr_TransMax := n4,

                                                  template (omit) CarrierFreqEUTRA      p_CarrierFreqEutra := omit )  runs on EUTRA_PTC

  { /* @sic R5s100491: enhancement to support MeasConfig sic@ */

    /* @sic R5s100524: common change of HO functions sic@ */

    /* @sic R5-106063: Sequence of inter-cell handover sic@ */

    /* @sic R5s100801: No CNF for SS configuration of steps 10..14 sic@

     *                 when there is data pending at the UE the UE may send this data immediately after the HO;

     *                 this would collide with CNFs sent during reconfigurations of source and target call after the HO;

     *                 on the other hand none of the reconfigurations in source or target cell should have impact in cases the test case sends data after the HO

     *                 => reconfiguration steps 10..14 are done without request of CNF as proposed in the CR */

    /* @sic R5s100772: p_UL_GrantConfig_SourceCell sic@ */

    /* @sic R5s110168: additional parameter p_Dsr_TransMax sic@ */

    /* @sic R5-112122: UL_DL_Earfcn is omit in case of inter-cell HO when there is no change of frequency sic@ */

    /* @sic R5s110279 change 1: new optional parameter p_CarrierFreqEutra to force the CarrierFreqEutra e.g. to omit the ul-CarrierFreq as requested for TC_8_2_4_9 sic@ */

    var DRB_IdentityList_Type v_DRB_IdentityList;

    var integer k;

    var EUTRA_SecurityParams_Type v_Auth_Params := f_EUTRA_Security_Get();

    var NextHopChainingCount v_NextHopChainingCount := 0; //According to 36.508 cl. 4.6.1 RRCConnectionReestablishment

    var Frequency_IE_Type v_Frequency_IE_SourceCell := f_EUTRA_CellInfo_GetFrequencyIEs (p_SourceCellId);

    var Frequency_IE_Type v_Frequency_IE_TargetCell := f_EUTRA_CellInfo_GetFrequencyIEs (p_TargetCellId);

    var CarrierFreqEUTRA v_CarrierFreqEutra_SourceCell := v_Frequency_IE_SourceCell.UL_DL_Earfcn;

    var CarrierFreqEUTRA v_CarrierFreqEutra_TargetCell := v_Frequency_IE_TargetCell.UL_DL_Earfcn;

    var template (omit) CarrierFreqEUTRA v_CarrierFreqEutra := omit;

    var template (value) MobilityControlInfo v_MobilityControlInfo;

    var template (value) RadioResourceConfigDedicated v_RadioResourceConfigDedicated := f_Generate_cs_508_RadioResourceConfigDedicated_HO (p_TargetCellId, p_Dsr_TransMax); 
    var template (value) MeasGapCtrl_Type v_MeasGapCtrl;

    var DrxCtrl_Type v_DrxCtrl;

    var boolean v_CnfFlag := tsc_NoCnfReq;  // @sic R5s100801: could be changed to an optional parameter if necessary sic@

    if (isvalue(p_CarrierFreqEutra)) {      // use the give CarrierFreqEutra; @sic R5s110279 change 1 sic@

      v_CarrierFreqEutra := p_CarrierFreqEutra;

    }

    else {                                  // => determine the CarrierFreqEutra depending on cell information stored for sourece and target cell

      if (v_CarrierFreqEutra_SourceCell != v_CarrierFreqEutra_TargetCell) {    // only if CarrierFreqEUTRA is different in both cells the CarrierFreqEUTRA of the target cell is handed over @sic R5-112122 sic@

        v_CarrierFreqEutra := v_CarrierFreqEutra_TargetCell;

      }

    }

    v_MobilityControlInfo := f_Generate_cs_MobilityControlInfo_HO_CRNTI(p_TargetCellId, v_CarrierFreqEutra, p_C_RNTI);

    v_RadioResourceConfigDedicated := f_Generate_cs_508_RadioResourceConfigDedicated_HO (p_TargetCellId, p_Dsr_TransMax);

    // Build DRB identity list from inputed radiobearer config list

    for (k:=0; k < lengthof(p_DrbConfigList); k:= k+1) {

      v_DRB_IdentityList[k] := valueof ( p_DrbConfigList[k].Id.Drb );

    }

    // Step 2: Tell the SS about the HO    @sic R5w100212 sic@

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverInit_REQ(p_SourceCellId, p_TargetCellId)); // @sic R5s100556, R5s100560 sic@

    // Step 3: Configure C-RNTI based contention resolution in target cell

    f_EUTRA_SS_ConfigRachProcedure_HO(p_TargetCellId, cs_TimingInfo_Now, p_C_RNTI);

    // Save new C-RNTI in cell configuration for p_TargetCellId

    f_EUTRA_CellInfo_SetC_RNTI ( p_TargetCellId, p_C_RNTI );

    // Step 4: activate RRC security at SS target cell for SRB 1, 2 and 1 AM DRB

    v_Auth_Params := f_EUTRA_SS_AS_ActivateSecurity_HO_Reest ( p_TargetCellId,

                                                               v_Auth_Params,

                                                               v_NextHopChainingCount,

                                                               -,

                                                               v_DRB_IdentityList);

    f_EUTRA_Security_Set ( v_Auth_Params );

    // Step 5: configure DRX and MeasGap at the target cell (if necessary)

    if (f_EUTRA_CellInfo_DrxIsConfigured(p_SourceCellId)) {

      v_DrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_SourceCellId);

      f_EUTRA_SS_DrxCtrlConfig(p_TargetCellId, v_DrxCtrl);

    }

    if (f_EUTRA_CellInfo_MeasGapIsConfigured(p_SourceCellId)) {
      v_MeasGapCtrl := f_EUTRA_CellInfo_GetMeasGapCtrl(p_SourceCellId);
      f_EUTRA_SS_MeasGapCtrlConfig(p_TargetCellId, v_MeasGapCtrl);
    }
    // Step 6: stop periodic TA on source cell   @sic R5s100520 change 10 sic@

    f_EUTRA_SS_CommonCellConfig(p_SourceCellId, cas_ULGrantAllocation_REQ(p_SourceCellId, cs_TimingInfo_Now, -, cs_PUCCH_Synch_None, p_UL_GrantConfig_SourceCell));

    // Step 7: configure UL grant configuration on target cell    @sic R5s100515 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_ULGrantAllocation_REQ(p_TargetCellId, cs_TimingInfo_Now, p_DciUlInfo_TargetCell, cs_PUCCH_Synch_None, p_UL_GrantConfig_TargetCell));

    // Step 8: Transmit RRCConnectionReconfiguration/HO command to UE

    f_EUTRA_RRC_ConnReconfigHO_Common(p_SourceCellId,

                                      p_MeasConfig,

                                      v_MobilityControlInfo,

                                      v_RadioResourceConfigDedicated,

                                      cs_508SecurityConfigHO_IntraLTE);

    // Step 9:

    SRB.receive(car_SRB1_RrcPdu_IND(p_TargetCellId, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));       // @sic R5s100430  Change 2 sic@

    // Step 10: configure periodic TA on target cell    @sic R5s100525 change 1.3 sic@

    f_EUTRA_SS_CommonCellConfig(p_TargetCellId, cas_PUCCH_Synch_Config_REQ(p_TargetCellId, cs_TimingInfo_Now, cs_PUCCH_Synch_Auto_Def, tsc_NoCnfReq)); // @sic R5s100801 sic@

    // Step 11: Tell the SS about the HO    @sic R5w100212 sic @

    f_EUTRA_SS_PdcpHandoverCtrl(p_TargetCellId, cas_PdcpHandoverComplete_REQ(p_TargetCellId)); // @sic R5s100556, R5s100560 sic@

    // Step 12: configure RACH procedure for target cell back to default    @sic R5s100520 change 11 sic@

    f_EUTRA_SS_ConfigRachProcedure (p_TargetCellId,

                                    omit,

                                    cs_RachProcedureConfig_Def(omit, f_EUTRA_CellInfo_GetRAR_TA(p_TargetCellId), f_EUTRA_CellInfo_GetDL_ChBandwidth(p_TargetCellId)),

                                    -,

                                    v_CnfFlag); // @sic R5s100801 sic@

    // Step 13: Re-establish SRBs and DRB(s) in source cell

    f_EUTRA_SS_SRBs_DRBs_Release ( p_SourceCellId, -, v_DRB_IdentityList, v_CnfFlag); // @sic R5s100801 sic@

    f_EUTRA_SS_SRBs_DRBs_Config ( p_SourceCellId, -, p_DrbConfigList, v_CnfFlag); // @sic R5s100801 sic@

    // Step 14: Release DRX and MeasGap configuration

    f_EUTRA_SS_DrxCtrlConfig(p_SourceCellId, cs_DrxCtrl_None, -, v_CnfFlag);  // @sic R5s100801 sic@

    f_EUTRA_SS_MeasGapCtrlConfig(p_SourceCellId, cs_MeasGapCtrl_None, -, v_CnfFlag);
  }

f_EUTRA_RRC_ConnectionRelease_Local

  function f_EUTRA_RRC_ConnectionRelease_Local(CellId_Type p_CellId,

                                               SubFrameTiming_Type p_TimingAtT,

                                               template (value) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_Def) runs on EUTRA_PTC

  { /* NOTE: the function schedules all steps to release and reconfigure the SRBs and (default) DRB

     *       => this will in fact happen in the future */

    /* @sic R5-110803 change 13 sic@ */

    /* @sic R5s110634 MCC160 implementation of change 1.1: new parameter p_DrbConfigList sic@ */

    var SubFrameTiming_Type v_Timing := p_TimingAtT;

    var DRB_IdentityList_Type v_DRB_IdentityList;

    var integer k;

    // Build DRB identity list from inputed radiobearer config list

    for (k:=0; k < lengthof(p_DrbConfigList); k:= k+1) {

      v_DRB_IdentityList[k] := valueof ( p_DrbConfigList[k].Id.Drb );

    }

    /* Step 1 at T: disable the UL grant

    *               => it is not possible for the UE to send any ACK for the RRC Connecion Release)

    *               => UE has to wait until 60ms timer has timed out (-> 36.331 cl. 5.3.8.3) */

    f_EUTRA_StopULGrantTransmission(p_CellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));

    /* Step 2 at T+5ms: release security */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 5);

    f_EUTRA_SS_RRC_ReleaseSecurity(p_CellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));

    /* Step 3 at T+10ms: release DRX */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 10);

    f_EUTRA_SS_DrxCtrlConfig(p_CellId, cs_DrxCtrl_None, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));

    /* Step 3A at T+15ms: release MeasGapConfig */
    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 15);
    f_EUTRA_SS_MeasGapCtrlConfig(p_CellId, cs_MeasGapConfig_None, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));
    /* Step 4 (obsolete) */

    /* Step 5 at T+55ms: Release SRBs and DRBs */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 55);

    f_EUTRA_SS_SRBs_DRBs_Release(p_CellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number), v_DRB_IdentityList);

    /* Step 6 at T+60ms: Configure SRBs and DRBs */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 60);

    f_EUTRA_SS_SRBs_DRBs_Config(p_CellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number), p_DrbConfigList);

  }

